
	Other comments:
	


Change 1
	Function name
	Type HTTP_ConfigurationParams

	Reason for change
	Following the implementation of change 1 of TTCN CR R5s160442 the length of the XRES to be used as the password for GBA authentication is calculated as lengthof (v_Xres)/8 where v_Xres= p_HTTP_AuthenticationParams.AuthParams.XRES. This will give the length of the XRES parameter calculated in f_AutehnticationInit() which will always give a length of 16 bytes. But the actual length of XRES to be used as the password for calculating the digest response is the length of the response given by the UE  during the previous AKA authentication (as already used for the SIP digest calculation). This value is currently known to EUTRA and IMS PTC but not HTTP_XCAP PTC.

	Summary of change
	Add a new field “xresLen” to HTTP configuration paramters

	TTCN module
	HTTP_XCAP_CoordMsgs.ttcn

	MCC160 Comment
	


Before change
  type record HTTP_ConfigurationParams_Type { /* @status    APPROVED (IMS) */

    IP_SocketList_Type XcapServerAddrAndPort_List,     /* @sic R5s150148 sic@ */

    IP_SocketList_Type BsfServerAddrAndPort_List,      /* @sic R5s150148 sic@ */

    IP_DrbInfo_Type DrbRoutingInfo,

    HTTP_AuthenticationMethod_Type AuthenticationMethod,

    charstring XcapUsername,                /* Username to be used for Authorisation */

    charstring XcapUserId,                  /* public user indentity to be used for XCAP signalling (XUI) */

    charstring XcapRootUri

  };
After change
  type record HTTP_ConfigurationParams_Type { /* @status    APPROVED (IMS) */

    IP_SocketList_Type XcapServerAddrAndPort_List,     /* @sic R5s150148 sic@ */

    IP_SocketList_Type BsfServerAddrAndPort_List,      /* @sic R5s150148 sic@ */

    IP_DrbInfo_Type DrbRoutingInfo,

    HTTP_AuthenticationMethod_Type AuthenticationMethod,

    charstring XcapUsername,                /* Username to be used for Authorisation */

    charstring XcapUserId,                  /* public user indentity to be used for XCAP signalling (XUI) */

    charstring XcapRootUri,

    integer xresLen
  };
Change 2
	Function name
	Template cs_HTTP_ConfigurationParamns

	Reason for change
	Following the implementation of change 1 of TTCN CR R5s160442 the length of the XRES to be used as the password for GBA authentication is calculated as lengthof (v_Xres)/8 where v_Xres= p_HTTP_AuthenticationParams.AuthParams.XRES. This will give the length of the XRES parameter calculated in f_AutehnticationInit() which will always give a length of 16 bytes. But the actual length of XRES to be used as the password for calculating the digest response is the length of the response given by the UE  during the previous AKA authentication (as already used for the SIP digest calculation). This value is currently known to EUTRA and IMS PTC but not HTTP_XCAP PTC.

	Summary of change
	Add a new field “xresLen” to HTTP configuration paramters

	TTCN module
	IMS_CC_XCAP_Testcases.ttcn

	MCC160 Comment
	


Before change
    template (value) HTTP_ConfigurationParams_Type cs_HTTP_ConfigurationParams(template (value) IP_SocketList_Type p_XcapServerAddrAndPort_List,

                                                                             template (value) IP_SocketList_Type p_BsfServerAddrAndPort_List,

                                                                             template (value) IP_DrbInfo_Type p_DrbRoutingInfo,

                                                                             HTTP_AuthenticationMethod_Type p_AuthenticationMethod,

                                                                             charstring p_XcapUsername,

                                                                             charstring p_XcapUserId,

                                                                             charstring p_XcapRootUri := px_XCAP_RootUri) :=

  { /* @status    APPROVED (IMS) */

    XcapServerAddrAndPort_List := p_XcapServerAddrAndPort_List,  // @sic R5s150148 sic@

    BsfServerAddrAndPort_List := p_BsfServerAddrAndPort_List,    // @sic R5s150148 sic@

    DrbRoutingInfo := p_DrbRoutingInfo,

    AuthenticationMethod := p_AuthenticationMethod,

    XcapUsername := p_XcapUsername,

    XcapUserId := p_XcapUserId,

    XcapRootUri := p_XcapRootUri

  };

After change

    template (value) HTTP_ConfigurationParams_Type cs_HTTP_ConfigurationParams(template (value) IP_SocketList_Type p_XcapServerAddrAndPort_List,

                                                                             template (value) IP_SocketList_Type p_BsfServerAddrAndPort_List,

                                                                             template (value) IP_DrbInfo_Type p_DrbRoutingInfo,

                                                                             HTTP_AuthenticationMethod_Type p_AuthenticationMethod,

                                                                             charstring p_XcapUsername,

                                                                             charstring p_XcapUserId,

                                                                             charstring p_XcapRootUri := px_XCAP_RootUri,

                                                                             integer p_XresLen := 8) :=

  { /* @status    APPROVED (IMS) */

    XcapServerAddrAndPort_List := p_XcapServerAddrAndPort_List,  // @sic R5s150148 sic@

    BsfServerAddrAndPort_List := p_BsfServerAddrAndPort_List,    // @sic R5s150148 sic@

    DrbRoutingInfo := p_DrbRoutingInfo,

    AuthenticationMethod := p_AuthenticationMethod,

    XcapUsername := p_XcapUsername,

    XcapUserId := p_XcapUserId,

    XcapRootUri := p_XcapRootUri,

    xresLen := p_XresLen 

  };
Change 3
	Function name
	F_IMS_HTTP_Configure()

	Reason for change
	Following the implementation of change 1 of TTCN CR R5s160442 the length of the XRES to be used as the password for GBA authentication is calculated as lengthof (v_Xres)/8 where v_Xres= p_HTTP_AuthenticationParams.AuthParams.XRES. This will give the length of the XRES parameter calculated in f_AutehnticationInit() which will always give a length of 16 bytes. But the actual length of XRES to be used as the password for calculating the digest response is the length of the response given by the UE  during the previous AKA authentication (as already used for the SIP digest calculation). This value is currently known to EUTRA and IMS PTC but not HTTP_XCAP PTC.

	Summary of change
	Set value of xresLen in HTTP configuration paramters from function f_IMS_PTC_Security_GetXRESLength()

	TTCN module
	IMS_CC_XCAP_Testcases.ttcn

	MCC160 Comment
	


Before change
  function f_IMS_HTTP_Configure(HTTP_AuthenticationMethod_Type p_AuthenticationMethod,

                                charstring p_XcapUserId,

                                charstring p_XcapUsername) runs on IMS_XCAP_PTC

  {

    var template (value) IMS_HTTP_Command_Type v_Command;

    var template (value) IP_AddrInfo_Type v_IP_AddressV4;

    var template (value) IP_AddrInfo_Type v_IP_AddressV6;

    var PDN_AddressInfo_Type v_PDN_AddressInfo := f_PDN_AddressInfo_Get(PDN_2); /* @sic R5s141051 change 8 sic@ */

    var template (value) IP_SocketList_Type v_XcapServerAddrAndPort_List := {};

    var template (value) IP_SocketList_Type v_BsfServerAddrAndPort_List := {};

    var template (value) IP_DrbInfo_Type v_DrbRoutingInfo;

    var template (omit)  IPCAN_RAN_Type v_RanType := f_IMS_PTC_GetRanType();   /* @sic R5-163201: IPCAN_RAN_Type instead of charstring sic@ */

    var IP_DrbId_Type v_DrbId;

    v_IP_AddressV4 := cs_IPv4Addr(v_PDN_AddressInfo.XCAP_ServerAddressIPv4);  /* @sic R5s150148 sic@ */

    v_IP_AddressV6 := cs_IPv6Addr(v_PDN_AddressInfo.XCAP_ServerAddressIPv6);  /* @sic R5s150148 sic@ */

    if (isvalue(v_RanType)) {

      select(valueof(v_RanType)) {

        case (EUTRA_FDD, EUTRA_TDD) {

          v_DrbId := f_EUTRA_EpsBearerAssociatedDRB(tsc_EpsDefaultBearerId2ndPDN);    /* DRB Id for 2nd PDN; @sic R5s141351 change 4 sic@ */

          v_DrbRoutingInfo := cs_DrbInfo_EUTRA(eutra_Cell1, v_DrbId);

        }

        case else {

          FatalError(__FILE__, __LINE__, "RAN type not supported by current implementation");

        }

      }

    } else {

      FatalError(__FILE__, __LINE__, "Unknown RAN type");

    }

    v_XcapServerAddrAndPort_List[0] := cs_IP_Socket(v_IP_AddressV4, tsc_PortNumberHTTP);   /* @sic R5s150148 sic@ */

    v_XcapServerAddrAndPort_List[1] := cs_IP_Socket(v_IP_AddressV6, tsc_PortNumberHTTP);   /* @sic R5s150148 sic@ */

    v_Command := {

      Configure := cs_HTTP_ConfigurationParams(v_XcapServerAddrAndPort_List,

                                               v_BsfServerAddrAndPort_List,

                                               v_DrbRoutingInfo,

                                               p_AuthenticationMethod,

                                               p_XcapUsername,

                                               p_XcapUserId)

    };

    f_IMS_HTTP_RunCommand(__FILE__, __LINE__, v_Command);

  }  
After change

      function f_IMS_HTTP_Configure(HTTP_AuthenticationMethod_Type p_AuthenticationMethod,

                                charstring p_XcapUserId,

                                charstring p_XcapUsername) runs on IMS_XCAP_PTC

  {

    var template (value) IMS_HTTP_Command_Type v_Command;

    var template (value) IP_AddrInfo_Type v_IP_AddressV4;

    var template (value) IP_AddrInfo_Type v_IP_AddressV6;

    var PDN_AddressInfo_Type v_PDN_AddressInfo := f_PDN_AddressInfo_Get(PDN_2); /* @sic R5s141051 change 8 sic@ */

    var template (value) IP_SocketList_Type v_XcapServerAddrAndPort_List := {};

    var template (value) IP_SocketList_Type v_BsfServerAddrAndPort_List := {};

    var template (value) IP_DrbInfo_Type v_DrbRoutingInfo;

    var template (omit)  IPCAN_RAN_Type v_RanType := f_IMS_PTC_GetRanType();   /* @sic R5-163201: IPCAN_RAN_Type instead of charstring sic@ */

    var IP_DrbId_Type v_DrbId;

    v_IP_AddressV4 := cs_IPv4Addr(v_PDN_AddressInfo.XCAP_ServerAddressIPv4);  /* @sic R5s150148 sic@ */

    v_IP_AddressV6 := cs_IPv6Addr(v_PDN_AddressInfo.XCAP_ServerAddressIPv6);  /* @sic R5s150148 sic@ */

    if (isvalue(v_RanType)) {

      select(valueof(v_RanType)) {

        case (EUTRA_FDD, EUTRA_TDD) {

          v_DrbId := f_EUTRA_EpsBearerAssociatedDRB(tsc_EpsDefaultBearerId2ndPDN);    /* DRB Id for 2nd PDN; @sic R5s141351 change 4 sic@ */

          v_DrbRoutingInfo := cs_DrbInfo_EUTRA(eutra_Cell1, v_DrbId);

        }

        case else {

          FatalError(__FILE__, __LINE__, "RAN type not supported by current implementation");

        }

      }

    } else {

      FatalError(__FILE__, __LINE__, "Unknown RAN type");

    }

    v_XcapServerAddrAndPort_List[0] := cs_IP_Socket(v_IP_AddressV4, tsc_PortNumberHTTP);   /* @sic R5s150148 sic@ */

    v_XcapServerAddrAndPort_List[1] := cs_IP_Socket(v_IP_AddressV6, tsc_PortNumberHTTP);   /* @sic R5s150148 sic@ */

    v_Command := {

      Configure := cs_HTTP_ConfigurationParams(v_XcapServerAddrAndPort_List,

                                               v_BsfServerAddrAndPort_List,

                                               v_DrbRoutingInfo,

                                               p_AuthenticationMethod,

                                               p_XcapUsername,

                                               p_XcapUserId,

                                               -,

                                               f_IMS_PTC_Security_GetXRESLength())
    };

    f_IMS_HTTP_RunCommand(__FILE__, __LINE__, v_Command);

  }
Change 4
	Function name
	f_HTTP_XCAP_Main()

	Reason for change
	Following the implementation of change 1 of TTCN CR R5s160442 the length of the XRES to be used as the password for GBA authentication is calculated as lengthof (v_Xres)/8 where v_Xres= p_HTTP_AuthenticationParams.AuthParams.XRES. This will give the length of the XRES parameter calculated in f_AutehnticationInit() which will always give a length of 16 bytes. But the actual length of XRES to be used as the password for calculating the digest response is the length of the response given by the UE  during the previous AKA authentication (as already used for the SIP digest calculation). This value is currently known to EUTRA and IMS PTC but not HTTP_XCAP PTC.

	Summary of change
	Pass XRES length received in HTTP configuration parametrers into f_HTTP_XCAP_Authentication()

	TTCN module
	HTTP_XCAP_Main.ttcn

	MCC160 Comment
	


Before change
    function f_HTTP_XCAP_Main() runs on HTTP_XCAP_PTC

  {

    var IMS_HTTP_Command_Type v_IMS_Command;

    var HTTP_ConfigurationParams_Type v_ConfigurationParams;

    var HTTP_AuthenticationMethod_Type v_AuthenticationMethod;

    var template (value) HTTP_AuthenticationResult_Type v_AuthenticationResult;

    var XCAP_RSP v_XcapResponse;

    var charstring v_SimservsDocument;

    var charstring v_DocSelector;

    var template (omit) charstring v_Error;

    var template (omit) HTTP_DATA_IND v_InitialHttpRequest;

    timer t_Watchdog;

    v_IMS_Command := f_HTTP_IMS_CoordMsg_ReceiveCommand(IMS);

    if (not ischosen(v_IMS_Command.Configure)) {

      FatalError(__FILE__, __LINE__, "invalid command");  // at the beginning HTTP needs to be configured

    }

    v_ConfigurationParams := v_IMS_Command.Configure;

    v_AuthenticationMethod := v_ConfigurationParams.AuthenticationMethod;

    // Configure HTTP server(s) -> IP_PTC

    f_HTTP_Configure(HTTPCTRL,

                     v_ConfigurationParams.XcapServerAddrAndPort_List,   // @sic R5s150148 sic@

                     v_ConfigurationParams.BsfServerAddrAndPort_List,    // @sic R5s150148 sic@

                     v_ConfigurationParams.DrbRoutingInfo);

    // Configure XCAP Server

    v_DocSelector := fl_DocSelector_Build(v_ConfigurationParams.XcapRootUri,

                                          v_ConfigurationParams.XcapUserId);

    f_XCAP_Initialise(XCAP, v_DocSelector);

    f_HTTP_IMS_CoordMsg_SendResponse(IMS);

    while (true) {

      v_IMS_Command := f_HTTP_IMS_CoordMsg_ReceiveCommand(IMS);     // Get command from IMS PTC

      if (ischosen(v_IMS_Command.DocumentSet)) {                    // Set document

        v_SimservsDocument := v_IMS_Command.DocumentSet;

        f_XCAP_Put(XCAP, fl_XcapUri_Build(v_DocSelector), tsc_Simservs_ContentType_Document, v_SimservsDocument);

        f_HTTP_IMS_CoordMsg_SendResponse(IMS);

      }

      else if (ischosen(v_IMS_Command.DocumentGet)) {               // Get document

        v_XcapResponse := f_XCAP_Get(XCAP, fl_XcapUri_Build(v_DocSelector));

        if (not ispresent(v_XcapResponse.xmlBody)) {

          FatalError(__FILE__, __LINE__, "File Not Found");

        } else {

          v_SimservsDocument := v_XcapResponse.xmlBody;

          f_HTTP_IMS_CoordMsg_SendResponse(IMS, v_SimservsDocument);

        }

      }

      else if (ischosen(v_IMS_Command.TransactionStart)) {           // Start transaction: HTTP requests and responses acc. to 34.229 C.29

        t_Watchdog.start(v_IMS_Command.TransactionStart.MaxDuration);

        v_InitialHttpRequest := omit;

        v_Error := omit;

        select (v_AuthenticationMethod) {

          case (httpDigestAuthentication, httpDigestAndGbaAuthentication) {

            // do authentication:

            v_AuthenticationResult := f_HTTP_XCAP_Authentication(HTTPDATA,

                                                                 t_Watchdog,

                                                                 v_AuthenticationMethod,

                                                                 v_ConfigurationParams.XcapUsername);

            if (ischosen(v_AuthenticationResult.Error)) {

              v_Error := v_AuthenticationResult.Error;

            }

            if (ischosen(v_AuthenticationResult.InitialRequest)) {

              v_InitialHttpRequest := v_AuthenticationResult.InitialRequest;

              v_Error := fl_HTTP_CheckRequestUri(valueof(v_InitialHttpRequest.httpRequest.requestLine.uri), v_DocSelector);

            }

            v_AuthenticationMethod := noAuthentication;   // no further authentication

          }

          case else {

            // no authentication

          }

        }

        if (not isvalue(v_Error)) {

          v_Error := fl_HTTP_SendReceiveLoop(t_Watchdog, v_IMS_Command.TransactionStart.InactivityDuration, v_DocSelector, v_InitialHttpRequest);

        }

        t_Watchdog.stop;

        f_HTTP_IMS_CoordMsg_SendResponse(IMS, -, v_Error);

      }

      else {

        FatalError(__FILE__, __LINE__, "invalid command");

      }

    }

  }

}
After change

  function f_HTTP_XCAP_Main() runs on HTTP_XCAP_PTC

  {

    var IMS_HTTP_Command_Type v_IMS_Command;

    var HTTP_ConfigurationParams_Type v_ConfigurationParams;

    var HTTP_AuthenticationMethod_Type v_AuthenticationMethod;

    var template (value) HTTP_AuthenticationResult_Type v_AuthenticationResult;

    var XCAP_RSP v_XcapResponse;

    var charstring v_SimservsDocument;

    var charstring v_DocSelector;

    var template (omit) charstring v_Error;

    var template (omit) HTTP_DATA_IND v_InitialHttpRequest;

    timer t_Watchdog;

    v_IMS_Command := f_HTTP_IMS_CoordMsg_ReceiveCommand(IMS);

    if (not ischosen(v_IMS_Command.Configure)) {

      FatalError(__FILE__, __LINE__, "invalid command");  // at the beginning HTTP needs to be configured

    }

    v_ConfigurationParams := v_IMS_Command.Configure;

    v_AuthenticationMethod := v_ConfigurationParams.AuthenticationMethod;

    // Configure HTTP server(s) -> IP_PTC

    f_HTTP_Configure(HTTPCTRL,

                     v_ConfigurationParams.XcapServerAddrAndPort_List,   // @sic R5s150148 sic@

                     v_ConfigurationParams.BsfServerAddrAndPort_List,    // @sic R5s150148 sic@

                     v_ConfigurationParams.DrbRoutingInfo);

    // Configure XCAP Server

    v_DocSelector := fl_DocSelector_Build(v_ConfigurationParams.XcapRootUri,

                                          v_ConfigurationParams.XcapUserId);

    f_XCAP_Initialise(XCAP, v_DocSelector);

    f_HTTP_IMS_CoordMsg_SendResponse(IMS);

    while (true) {

      v_IMS_Command := f_HTTP_IMS_CoordMsg_ReceiveCommand(IMS);     // Get command from IMS PTC

      if (ischosen(v_IMS_Command.DocumentSet)) {                    // Set document

        v_SimservsDocument := v_IMS_Command.DocumentSet;

        f_XCAP_Put(XCAP, fl_XcapUri_Build(v_DocSelector), tsc_Simservs_ContentType_Document, v_SimservsDocument);

        f_HTTP_IMS_CoordMsg_SendResponse(IMS);

      }

      else if (ischosen(v_IMS_Command.DocumentGet)) {               // Get document

        v_XcapResponse := f_XCAP_Get(XCAP, fl_XcapUri_Build(v_DocSelector));

        if (not ispresent(v_XcapResponse.xmlBody)) {

          FatalError(__FILE__, __LINE__, "File Not Found");

        } else {

          v_SimservsDocument := v_XcapResponse.xmlBody;

          f_HTTP_IMS_CoordMsg_SendResponse(IMS, v_SimservsDocument);

        }

      }

      else if (ischosen(v_IMS_Command.TransactionStart)) {           // Start transaction: HTTP requests and responses acc. to 34.229 C.29

        t_Watchdog.start(v_IMS_Command.TransactionStart.MaxDuration);

        v_InitialHttpRequest := omit;

        v_Error := omit;

        select (v_AuthenticationMethod) {

          case (httpDigestAuthentication, httpDigestAndGbaAuthentication) {

            // do authentication:

            v_AuthenticationResult := f_HTTP_XCAP_Authentication(HTTPDATA,

                                                                 t_Watchdog,

                                                                 v_AuthenticationMethod,

                                                                 v_ConfigurationParams.XcapUsername, 

  



















 v_ConfigurationParams.xresLen);

            if (ischosen(v_AuthenticationResult.Error)) {

              v_Error := v_AuthenticationResult.Error;

            }

            if (ischosen(v_AuthenticationResult.InitialRequest)) {

              v_InitialHttpRequest := v_AuthenticationResult.InitialRequest;

              v_Error := fl_HTTP_CheckRequestUri(valueof(v_InitialHttpRequest.httpRequest.requestLine.uri), v_DocSelector);

            }

            v_AuthenticationMethod := noAuthentication;   // no further authentication

          }

          case else {

            // no authentication

          }

        }

        if (not isvalue(v_Error)) {

          v_Error := fl_HTTP_SendReceiveLoop(t_Watchdog, v_IMS_Command.TransactionStart.InactivityDuration, v_DocSelector, v_InitialHttpRequest);

        }

        t_Watchdog.stop;

        f_HTTP_IMS_CoordMsg_SendResponse(IMS, -, v_Error);

      }

      else {

        FatalError(__FILE__, __LINE__, "invalid command");

      }

    }

  }

}
Change 5
	Function name
	f_HTTP_XCAP_Authentication()

	Reason for change
	Following the implementation of change 1 of TTCN CR R5s160442 the length of the XRES to be used as the password for GBA authentication is calculated as lengthof (v_Xres)/8 where v_Xres= p_HTTP_AuthenticationParams.AuthParams.XRES. This will give the length of the XRES parameter calculated in f_AutehnticationInit() which will always give a length of 16 bytes. But the actual length of XRES to be used as the password for calculating the digest response is the length of the response given by the UE  during the previous AKA authentication (as already used for the SIP digest calculation). This value is currently known to EUTRA and IMS PTC but not HTTP_XCAP PTC.

	Summary of change
	Use XRES length passed into the function for GBA authentication

	TTCN module
	HTTP_XCAP_Main.ttcn

	MCC160 Comment
	


Before change
  function f_HTTP_XCAP_Authentication(HTTP_DATA_PORT p_HttpDataPort,

                                      timer p_Watchdog,

                                      HTTP_AuthenticationMethod_Type p_AuthenticationMethod,

                                      charstring p_XcapUsername) return template (value) HTTP_AuthenticationResult_Type

  { /* @sic R5-151795, R5s150945: restructuring of authentication procedure sic@ */

    /* @sic R5-160765: simplification, generic implementation for XCAP and ProSe sic@ */

    var template (omit) HTTP_DATA_IND v_HTTP_DATA_IND;

    var HTTP_DATA_IND v_InitialRequestXCAP;

    var HTTP_DATA_IND v_SubsequentRequestXCAP;

    var charstring v_UsernameXCAP := p_XcapUsername;

    var charstring v_PasswordXCAP;

    var charstring v_RealmXCAP;

    var bitstring v_XRES;

    var integer v_XRESLength;

    var HTTP_AuthenticationParams_Type v_HTTP_AuthenticationParams := f_HTTP_AuthenticationInit();

    var ServerVal_List v_ProductTokenList;

    var template (value) WwwAuthenticate v_WwwAuthenticate;

    var template UserAgent v_UserAgent;

    var template Authorization v_Authorization;

    var boolean v_GbaAuthenticationRequired := (p_AuthenticationMethod == httpDigestAndGbaAuthentication);

    var template (value) HTTP_AuthenticationResult_Type v_AuthenticationResult;

    var charstring v_Error := "";

    if (v_GbaAuthenticationRequired) {

      v_RealmXCAP := "3GPP-bootstrapping@" & v_HTTP_AuthenticationParams.HomeDomainName;    /* @sic R5s150945 change 1.7 sic@ */

    }

    else {

      v_RealmXCAP := v_HTTP_AuthenticationParams.HomeDomainName;

    }

    // 29.1 step 2

    v_UserAgent := fl_UserAgentRX_XCAP();

    v_Authorization := *;

    v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, cr_HttpRequestXCAP(v_UserAgent, v_Authorization), p_Watchdog);  /* @sic R5-153795: support of ProSe - cr_HttpRequest -> cr_HttpRequestXCAP sic@ */

    if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

    v_InitialRequestXCAP := valueof(v_HTTP_DATA_IND);

    // 29.1 step 3a

    v_ProductTokenList := {"XCAP-Server"};

    v_WwwAuthenticate := fl_WwwAuthenticate_XCAP(v_RealmXCAP, v_HTTP_AuthenticationParams.Nonce);

    f_HTTP_SendResponse(p_HttpDataPort, v_InitialRequestXCAP.routingInfo, cs_HttpResponse_401(v_ProductTokenList, v_WwwAuthenticate));

    if (v_GbaAuthenticationRequired) {    // GBA authentication

      v_AuthenticationResult := f_HTTP_GBA_Authentication(p_HttpDataPort,

                                                          p_Watchdog,

                                                          NAF_XCAP,

                                                          v_HTTP_AuthenticationParams);

      if (not ischosen(v_AuthenticationResult.NoError)) { return v_AuthenticationResult; }

      v_UsernameXCAP := v_HTTP_AuthenticationParams.PSK_BootstrappingInfo.BTid;

      v_XRES := v_HTTP_AuthenticationParams.PSK_BootstrappingInfo.Ks_NAF;

      v_XRESLength := lengthof(v_XRES) / 8;

    } else {

      v_PasswordXCAP := "xcap";    /* @sic R5s130794 Change 25 sic@

                                      @sic R5-151795: evaluation of the password moved from f_IMS_HTTP_Transaction (IMS_XCAP_PTC) to here sic@ */

      v_XRES := oct2bit(char2oct(v_PasswordXCAP));

      v_XRESLength := lengthof(v_PasswordXCAP);

    }

    // 29.1 step 3b

    v_UserAgent := fl_UserAgentRX_XCAP();

    v_Authorization := cr_AuthorizationCommon(v_UsernameXCAP, v_RealmXCAP, v_HTTP_AuthenticationParams.Nonce, v_InitialRequestXCAP.httpRequest.requestLine.uri, tsc_HTTP_MD5);

    v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, cr_HttpRequestXCAP(v_UserAgent, v_Authorization), p_Watchdog);  /* @sic R5-153795: support of ProSe - cr_HttpRequest -> cr_HttpRequestXCAP sic@ */

    if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

    v_SubsequentRequestXCAP := valueof(v_HTTP_DATA_IND);

    // Check authentication response

    v_Error := f_HTTP_CheckDigestResponse("XCAP-Server", v_SubsequentRequestXCAP.httpRequest, v_XRES, v_XRESLength);

    if (v_Error != "") {

      return cs_HTTP_AuthenticationResult_Error(v_Error);

    }

    return cs_HTTP_AuthenticationResult_InitialRequest(v_SubsequentRequestXCAP);

  }
After change

    function f_HTTP_XCAP_Authentication(HTTP_DATA_PORT p_HttpDataPort,

                                      timer p_Watchdog,

                                      HTTP_AuthenticationMethod_Type p_AuthenticationMethod,

                                      charstring p_XcapUsername, integer p_XresLen) return template (value) HTTP_AuthenticationResult_Type

  { /* @sic R5-151795, R5s150945: restructuring of authentication procedure sic@ */

    /* @sic R5-160765: simplification, generic implementation for XCAP and ProSe sic@ */

    var template (omit) HTTP_DATA_IND v_HTTP_DATA_IND;

    var HTTP_DATA_IND v_InitialRequestXCAP;

    var HTTP_DATA_IND v_SubsequentRequestXCAP;

    var charstring v_UsernameXCAP := p_XcapUsername;

    var charstring v_PasswordXCAP;

    var charstring v_RealmXCAP;

    var bitstring v_XRES;

    var integer v_XRESLength;

    var HTTP_AuthenticationParams_Type v_HTTP_AuthenticationParams := f_HTTP_AuthenticationInit();

    var ServerVal_List v_ProductTokenList;

    var template (value) WwwAuthenticate v_WwwAuthenticate;

    var template UserAgent v_UserAgent;

    var template Authorization v_Authorization;

    var boolean v_GbaAuthenticationRequired := (p_AuthenticationMethod == httpDigestAndGbaAuthentication);

    var template (value) HTTP_AuthenticationResult_Type v_AuthenticationResult;

    var charstring v_Error := "";

    if (v_GbaAuthenticationRequired) {

      v_RealmXCAP := "3GPP-bootstrapping@" & v_HTTP_AuthenticationParams.HomeDomainName;    /* @sic R5s150945 change 1.7 sic@ */

    }

    else {

      v_RealmXCAP := v_HTTP_AuthenticationParams.HomeDomainName;

    }

    // 29.1 step 2

    v_UserAgent := fl_UserAgentRX_XCAP();

    v_Authorization := *;

    v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, cr_HttpRequestXCAP(v_UserAgent, v_Authorization), p_Watchdog);  /* @sic R5-153795: support of ProSe - cr_HttpRequest -> cr_HttpRequestXCAP sic@ */

    if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

    v_InitialRequestXCAP := valueof(v_HTTP_DATA_IND);

    // 29.1 step 3a

    v_ProductTokenList := {"XCAP-Server"};

    v_WwwAuthenticate := fl_WwwAuthenticate_XCAP(v_RealmXCAP, v_HTTP_AuthenticationParams.Nonce);

    f_HTTP_SendResponse(p_HttpDataPort, v_InitialRequestXCAP.routingInfo, cs_HttpResponse_401(v_ProductTokenList, v_WwwAuthenticate));

    if (v_GbaAuthenticationRequired) {    // GBA authentication

      v_AuthenticationResult := f_HTTP_GBA_Authentication(p_HttpDataPort,

                                                          p_Watchdog,

                                                          NAF_XCAP,

                                                          v_HTTP_AuthenticationParams, p_XresLen);

      if (not ischosen(v_AuthenticationResult.NoError)) { return v_AuthenticationResult; }

      v_UsernameXCAP := v_HTTP_AuthenticationParams.PSK_BootstrappingInfo.BTid;

      v_XRES := v_HTTP_AuthenticationParams.PSK_BootstrappingInfo.Ks_NAF;

      v_XRESLength := lengthof(v_XRES) / 8;

    } else {

      v_PasswordXCAP := "xcap";    /* @sic R5s130794 Change 25 sic@

                                      @sic R5-151795: evaluation of the password moved from f_IMS_HTTP_Transaction (IMS_XCAP_PTC) to here sic@ */

      v_XRES := oct2bit(char2oct(v_PasswordXCAP));

      v_XRESLength := lengthof(v_PasswordXCAP);

    }

    // 29.1 step 3b

    v_UserAgent := fl_UserAgentRX_XCAP();

    v_Authorization := cr_AuthorizationCommon(v_UsernameXCAP, v_RealmXCAP, v_HTTP_AuthenticationParams.Nonce, v_InitialRequestXCAP.httpRequest.requestLine.uri, tsc_HTTP_MD5);

    v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, cr_HttpRequestXCAP(v_UserAgent, v_Authorization), p_Watchdog);  /* @sic R5-153795: support of ProSe - cr_HttpRequest -> cr_HttpRequestXCAP sic@ */

    if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

    v_SubsequentRequestXCAP := valueof(v_HTTP_DATA_IND);

    // Check authentication response

    v_Error := f_HTTP_CheckDigestResponse("XCAP-Server", v_SubsequentRequestXCAP.httpRequest, v_XRES, v_XRESLength);

    if (v_Error != "") {

      return cs_HTTP_AuthenticationResult_Error(v_Error);

    }

    return cs_HTTP_AuthenticationResult_InitialRequest(v_SubsequentRequestXCAP);

  }
Change 6
	Function name
	f_HTTP_XCAP_Main()

	Reason for change
	Following the implementation of change 1 of TTCN CR R5s160442 the length of the XRES to be used as the password for GBA authentication is calculated as lengthof (v_Xres)/8 where v_Xres= p_HTTP_AuthenticationParams.AuthParams.XRES. This will give the length of the XRES parameter calculated in f_AutehnticationInit() which will always give a length of 16 bytes. But the actual length of XRES to be used as the password for calculating the digest response is the length of the response given by the UE  during the previous AKA authentication (as already used for the SIP digest calculation). This value is currently known to EUTRA and IMS PTC but not HTTP_XCAP PTC.

	Summary of change
	Use XRES length passed into the function for digest response calculation

	TTCN module
	HTTP_CommonAuthentication.ttcn

	MCC160 Comment
	


Before change
  function f_HTTP_GBA_Authentication(HTTP_DATA_PORT p_HttpDataPort,

                                     timer p_Watchdog,

                                     NAF_Type p_NAF,

                                     HTTP_AuthenticationParams_Type p_HTTP_AuthenticationParams) return template (value) HTTP_AuthenticationResult_Type

  { /* @sic R5-160765: generic implementation for XCAP and ProSe sic@ */

    var template (omit) HTTP_DATA_IND v_HTTP_DATA_IND;

    var HTTP_DATA_IND v_InitialRequestGBA;

    var HTTP_DATA_IND v_SubsequentRequestGBA;

    var charstring v_RequestLineUri;

    var charstring v_UsernameGBA;

    var bitstring v_XRES := p_HTTP_AuthenticationParams.AuthParams.XRES;

    var integer v_XRESLength := lengthof(v_XRES) / 8;  /* @sic R5s160442 change 1 sic@ */

    var charstring v_CNonce;

    var ServerVal_List v_ProductTokenList;

    var template (value) WwwAuthenticate v_WwwAuthenticate;

    var template (value) AuthenticationInfo v_AuthenticationInfo;

    var template (value) ContentType v_ContentType;

    var charstring v_MessageBody;

    var template (present) HttpRequest_Type v_HttpRequest;

    var template (present) charstring v_RequestLineMethod;

    var template UserAgent v_UserAgent;

    var template Authorization v_Authorization;

    var template (present) Host v_Host;

    var octetstring v_DigestResponse;

    var charstring v_RspAuth;

    var boolean v_GbaInitialRequest := true;

    var charstring v_Error := "";

    v_Host := cr_Host(p_HTTP_AuthenticationParams.BsfServerDomainName);     /* @sic R5-155363 sic@ */

    if (pc_TMPI_forGBA) {

      v_UsernameGBA := fl_GBA_TMPI_Derivation(p_HTTP_AuthenticationParams.AuthParams, p_HTTP_AuthenticationParams.PrivateUserId, p_HTTP_AuthenticationParams.BsfServerDomainName);

    } else {

      v_UsernameGBA := p_HTTP_AuthenticationParams.PrivateUserId;

    }

    select (p_NAF) {

      case (NAF_XCAP, NAF_ProSe) {      /* @sic R5-160765: GBA for ProSE; !!!! most likely for ProSe a different method than GET needs to be used !!!! sic@ */

        v_RequestLineMethod := "GET";

      }

    }

    // 29.2 step 1

    v_ProductTokenList := {"3gpp-gba-tmpi"};

    v_UserAgent := f_UserAgentRX(v_ProductTokenList);

    v_Authorization := fl_Authorization_GBA(v_UsernameGBA, p_HTTP_AuthenticationParams.BsfServerDomainName, v_GbaInitialRequest);

    v_HttpRequest := cr_HttpRequest(cr_HttpRequestLine(v_RequestLineMethod), v_UserAgent, v_Authorization, v_Host); /* @sic R5-160765: GBA for ProSE; R5-155363: v_Host sic@ */

    v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, v_HttpRequest, p_Watchdog);

    if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

    v_InitialRequestGBA := valueof(v_HTTP_DATA_IND);

    /* Authentication is based on standard 3GPP AKA infratructure which sets the password (XRES)

       used in the HTTP Digest authentication according RFC 3310.

       See TS 33.220 clause 4.3.1. */

    // 29.2 step 2

    v_ProductTokenList := {"BSF-Server"};

    v_WwwAuthenticate := fl_WwwAuthenticate_GBA(p_HTTP_AuthenticationParams.Nonce, p_HTTP_AuthenticationParams.BsfServerDomainName); /* @sic R5s160442 change 2: p_HTTP_AuthenticationParams.BsfServerDomainName sic@ */

    f_HTTP_SendResponse(p_HttpDataPort, v_InitialRequestGBA.routingInfo, cs_HttpResponse_401(v_ProductTokenList, v_WwwAuthenticate));

    // 29.2 step 3

    v_RequestLineUri := v_InitialRequestGBA.httpRequest.requestLine.uri;    // same URI as in previous request */

    v_UserAgent := *;

    v_Authorization := fl_Authorization_GBA(v_UsernameGBA, p_HTTP_AuthenticationParams.BsfServerDomainName);

    v_HttpRequest := cr_HttpRequest(cr_HttpRequestLine(v_RequestLineMethod, v_RequestLineUri), v_UserAgent, v_Authorization, v_Host); /* @sic R5-160765: GBA for ProSE; R5-155363: v_Host sic@ */

    v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, v_HttpRequest, p_Watchdog);

    if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

    v_SubsequentRequestGBA := valueof(v_HTTP_DATA_IND);

    // Check authentication response

    v_Error := f_HTTP_CheckDigestResponse("BSF-Server", v_SubsequentRequestGBA.httpRequest, v_XRES, v_XRESLength);

    if (v_Error != "") { return cs_HTTP_AuthenticationResult_Error(v_Error); }

    v_CNonce := f_SIP_CommaParamList_GetParamValue(v_SubsequentRequestGBA.httpRequest.authorization.body[0].digestResponse, "cnonce");

    // 29.2 step 4

    v_ProductTokenList := {"3gpp-gba-tmpi"};  /* @sic R5s150945 change 1.5: "3gpp-gba-tmpi" only sic@ */

    v_ContentType := cs_ContentType("application/vnd.3gpp.bsf+xml");

    v_MessageBody := oct2char(bit2oct(encvalue(cs_BootstrappingInfo(p_HTTP_AuthenticationParams.PSK_BootstrappingInfo.BTid))));

    v_DigestResponse := f_IMS_Authorization_CalculateDigestResponse("", v_SubsequentRequestGBA.httpRequest.authorization.body[0].digestResponse, v_XRES, v_XRESLength, v_MessageBody); /* @sic R5s150945 change 1.6 sic@ */

    v_RspAuth := f_StringToLower(oct2str(v_DigestResponse));

    v_AuthenticationInfo := fl_AuthenticationInfo_GBA(v_RspAuth, v_CNonce);

    f_HTTP_SendResponse(p_HttpDataPort,

                        v_SubsequentRequestGBA.routingInfo,

                        cs_HttpResponse_200(-, cs_Server(v_ProductTokenList), v_AuthenticationInfo, v_ContentType, v_MessageBody));

    return cs_HTTP_AuthenticationResult_NoError;

  }

}

After change

  function f_HTTP_GBA_Authentication(HTTP_DATA_PORT p_HttpDataPort,

                                     timer p_Watchdog,

                                     NAF_Type p_NAF,

                                     HTTP_AuthenticationParams_Type p_HTTP_AuthenticationParams, integer p_XresLen)

) return template (value) HTTP_AuthenticationResult_Type

  { /* @sic R5-160765: generic implementation for XCAP and ProSe sic@ */

    var template (omit) HTTP_DATA_IND v_HTTP_DATA_IND;

    var HTTP_DATA_IND v_InitialRequestGBA;

    var HTTP_DATA_IND v_SubsequentRequestGBA;

    var charstring v_RequestLineUri;

    var charstring v_UsernameGBA;

    var bitstring v_XRES := p_HTTP_AuthenticationParams.AuthParams.XRES;

    var integer v_XRESLength := p_XresLen;
    var charstring v_CNonce;

    var ServerVal_List v_ProductTokenList;

    var template (value) WwwAuthenticate v_WwwAuthenticate;

    var template (value) AuthenticationInfo v_AuthenticationInfo;

    var template (value) ContentType v_ContentType;

    var charstring v_MessageBody;

    var template (present) HttpRequest_Type v_HttpRequest;

    var template (present) charstring v_RequestLineMethod;

    var template UserAgent v_UserAgent;

    var template Authorization v_Authorization;

    var template (present) Host v_Host;

    var octetstring v_DigestResponse;

    var charstring v_RspAuth;

    var boolean v_GbaInitialRequest := true;

    var charstring v_Error := "";

    v_Host := cr_Host(p_HTTP_AuthenticationParams.BsfServerDomainName);     /* @sic R5-155363 sic@ */

    if (pc_TMPI_forGBA) {

      v_UsernameGBA := fl_GBA_TMPI_Derivation(p_HTTP_AuthenticationParams.AuthParams, p_HTTP_AuthenticationParams.PrivateUserId, p_HTTP_AuthenticationParams.BsfServerDomainName);

    } else {

      v_UsernameGBA := p_HTTP_AuthenticationParams.PrivateUserId;

    }

    select (p_NAF) {

      case (NAF_XCAP, NAF_ProSe) {      /* @sic R5-160765: GBA for ProSE; !!!! most likely for ProSe a different method than GET needs to be used !!!! sic@ */

        v_RequestLineMethod := "GET";

      }

    }

    // 29.2 step 1

    v_ProductTokenList := {"3gpp-gba-tmpi"};

    v_UserAgent := f_UserAgentRX(v_ProductTokenList);

    v_Authorization := fl_Authorization_GBA(v_UsernameGBA, p_HTTP_AuthenticationParams.BsfServerDomainName, v_GbaInitialRequest);

    v_HttpRequest := cr_HttpRequest(cr_HttpRequestLine(v_RequestLineMethod), v_UserAgent, v_Authorization, v_Host); /* @sic R5-160765: GBA for ProSE; R5-155363: v_Host sic@ */

    v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, v_HttpRequest, p_Watchdog);

    if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

    v_InitialRequestGBA := valueof(v_HTTP_DATA_IND);

    /* Authentication is based on standard 3GPP AKA infratructure which sets the password (XRES)

       used in the HTTP Digest authentication according RFC 3310.

       See TS 33.220 clause 4.3.1. */

    // 29.2 step 2

    v_ProductTokenList := {"BSF-Server"};

    v_WwwAuthenticate := fl_WwwAuthenticate_GBA(p_HTTP_AuthenticationParams.Nonce, p_HTTP_AuthenticationParams.BsfServerDomainName); /* @sic R5s160442 change 2: p_HTTP_AuthenticationParams.BsfServerDomainName sic@ */

    f_HTTP_SendResponse(p_HttpDataPort, v_InitialRequestGBA.routingInfo, cs_HttpResponse_401(v_ProductTokenList, v_WwwAuthenticate));

    // 29.2 step 3

    v_RequestLineUri := v_InitialRequestGBA.httpRequest.requestLine.uri;    // same URI as in previous request */

    v_UserAgent := *;

    v_Authorization := fl_Authorization_GBA(v_UsernameGBA, p_HTTP_AuthenticationParams.BsfServerDomainName);

    v_HttpRequest := cr_HttpRequest(cr_HttpRequestLine(v_RequestLineMethod, v_RequestLineUri), v_UserAgent, v_Authorization, v_Host); /* @sic R5-160765: GBA for ProSE; R5-155363: v_Host sic@ */

    v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, v_HttpRequest, p_Watchdog);

    if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

    v_SubsequentRequestGBA := valueof(v_HTTP_DATA_IND);

    // Check authentication response

    v_Error := f_HTTP_CheckDigestResponse("BSF-Server", v_SubsequentRequestGBA.httpRequest, v_XRES, v_XRESLength);

    if (v_Error != "") { return cs_HTTP_AuthenticationResult_Error(v_Error); }

    v_CNonce := f_SIP_CommaParamList_GetParamValue(v_SubsequentRequestGBA.httpRequest.authorization.body[0].digestResponse, "cnonce");

    // 29.2 step 4

    v_ProductTokenList := {"3gpp-gba-tmpi"};  /* @sic R5s150945 change 1.5: "3gpp-gba-tmpi" only sic@ */

    v_ContentType := cs_ContentType("application/vnd.3gpp.bsf+xml");

    v_MessageBody := oct2char(bit2oct(encvalue(cs_BootstrappingInfo(p_HTTP_AuthenticationParams.PSK_BootstrappingInfo.BTid))));

    v_DigestResponse := f_IMS_Authorization_CalculateDigestResponse("", v_SubsequentRequestGBA.httpRequest.authorization.body[0].digestResponse, v_XRES, v_XRESLength, v_MessageBody); /* @sic R5s150945 change 1.6 sic@ */

    v_RspAuth := f_StringToLower(oct2str(v_DigestResponse));

    v_AuthenticationInfo := fl_AuthenticationInfo_GBA(v_RspAuth, v_CNonce);

    f_HTTP_SendResponse(p_HttpDataPort,

                        v_SubsequentRequestGBA.routingInfo,

                        cs_HttpResponse_200(-, cs_Server(v_ProductTokenList), v_AuthenticationInfo, v_ContentType, v_MessageBody));

    return cs_HTTP_AuthenticationResult_NoError;

  }

}

